

バルブ損失水頭を考慮した節点法による管網計算法

河 村 明 小 谷 辰 士

九州大学工学部水工土木学科助手 工博 九州大学大学院工学研究科修士課程在学中

上 田 年 比 古 神 野 健 二

九州大学工学部水工土木学科教授 工博 九州大学工学部水工土木学科助教授 工博

1. はじめに

近年我が国の都市上水道では、その配水システムが大規模・複雑化するに伴い、漏水量の抑制と需要者への供給水圧の確保を目的とした圧力制御の問題が重要な課題となっている。これには、配水管網内の圧力制御用バルブにより水圧管理を行う必要がある。福岡市では昭和53年の大湯水をきっかけとして配水調整事業に取り組み、現在では市内配水管網の各要所に流量計47個、水圧計99個、電動調整バルブ 123 個を設置している。そして、本局の中央監視装置で流量計、水圧計から時々刻々得られるセンサ情報に基づいて給水の状況を把握し、遠隔操作による電動調整バルブの作動により余剰水圧の抑制や各浄水場相互の流量調節を行っている¹⁾。このような配水システムの解析²⁾にあたっては、摩擦損失水頭のみならず、バルブによる形状損失水頭を考慮した管網計算が必要となる。この場合、バルブ損失水頭式は流量の2乗で表現されているが、摩擦損失水頭式は採用する平均流速公式により式中の流量の次数が異なっている³⁾、上水道の管網計算に一般に用いられている Hazen-Williams の式(以下H-W式と記す)では流量の1.85乗で表現されている。このように両損失水頭式の流量の次数が異なる場合の管網計算としてはバルブ設置点で管路を分割し、バルブの両端に新たな仮想節点を設けて節点数を増やし、通常の管網計算アルゴリズムに帰着させて計算を行う方法が考えられる^{4),5)}。しかし、この方法では管網内のバルブの配置や設置数を何度も変更する配水シミュレーションにおいては、隣接節点情報等の入力データ値の変更にかかり手間がかかり、また計算機容量が増大する。

本報では、摩擦損失水頭及びバルブ損失水頭を含む管網において、バルブの両端に新たな仮想節点を付加せず節点法^{6),7)}により管網計算を行う方法について検討している。すなわち、まず節点法による管網計算法を、管網内にバルブが設置された場合にも節点数を増加させるこ

となく適用できるように拡張し、その定式化を行う。次いで、入力データとして管路のバルブ存在の有無及びバルブ損失係数の情報を加えるだけで、節点数を増加させることなく管網内のバルブの有無に容易に対応できる管網計算プログラムを紹介し、実際の配水管網系を用いたシミュレーション結果を報告する。

2. バルブ損失がある場合の節点法による管網計算

2-1 概 説

節点法はエネルギー位法とも呼ばれ、各節点水頭を仮定して、これより得られる流量が各節点での連続条件を満足するまで水頭の補正計算を繰り返すものである^{6),7)}。節点法は各節点水頭と各管路流量が直接計算式中に表れるため、水圧制御への応用に非常に有利であり、また、初期仮定値にも特に工夫を要しない点で比較的取り扱いの容易な計算法である⁴⁾。

管網の基礎方程式は、任意の節点*i*における連続式(1)、任意の2節点*i*、*j*間の管路についての流量と水頭の関係式(2)である。ここでは水頭損失式としてH-W式を考えることにする。

$$\sum_j Q_{ij} = -q_i - Q_{i0} \dots \dots \dots (1)$$

$$H_i - H_j = r_{ij}^{-1/a} |Q_{ij}|^{(1/a)-1} Q_{ij} + F_{vij} |Q_{ij}| Q_{ij} \dots \dots \dots (2)$$

ここに、| | : 絶対値記号

Q_{ij} : 節点*i*から*j*に流れる管路流量 (m³/h) で $Q_{ij} = -Q_{ji}$

Q_{i0} : 節点*i*から管網系外部への流出量 (m³/h)(添字0は管網系外を指すものとする)。したがって、 Q_{0i} (外部から節点*i*への流入量) = $-Q_{i0}$

q_i : 節点*i*における取り出し水量 (m³/h)

H_i, H_j : 節点*i*、*j*における全水頭 (m) で、以下単に水頭と記す。

$$F_{vij} = 8f_{vij} / (g\pi^2 D_{ij}^4) (h^2/m^5) \dots\dots\dots(3)$$

f_{vij} : 節点 i, j 間に存在するバルブの損失係数

g : 重力加速度で, $g = 9.8 \times 3, 600^2 (m/h^2)$

α : 定数で, $\alpha = 0.54$

r_{ij} : 管路 ij の固有の定数であり,

$$r_{ij} = 0.27853 C_{ij} D_{ij}^{2.63} L_{ij}^{-0.54} (m^{2.46}/h) \dots\dots\dots(4)$$

C_{ij} : 節点 i, j 間の管路の流速係数 ($m^{0.87}/h$)

D_{ij} : 節点 i, j 間の管路の管径 (m)

L_{ij} : 節点 i, j 間の管路の管路長 (m)

いま管網の節点数 n_1 , 管路数 n_2 , 外部からの流入出量数 N とすれば, 一般の管網計算において既知量を, n_1 個の節点取り出し水量, 1 個の基準節点水頭及び $(N-1)$ 個の外部流入出量の合計 (n_1+N) 個とし, 未知量を (n_1-1) 個の節点水頭, n_2 個の管路流量及び 1 個の外部流入出量の合計 (n_1+n_2) 個とする。一方, 方程式の数は式(1)と式(2)の合計 (n_1+n_2) 個であるので未知量を求めることができる。さて節点法においては, 式(2)から得られる流量 Q_{ij} を式(1)の連続式に代入して Q_{ij} を消去すれば, 水頭に関する方程式が n_1 個得られる。この場合, 基準節点水頭を除いた (n_1-1) 個の節点水頭及び 1 個の外部流入出量が未知数となり, 方程式の数は n_1 個の連続式である。ここで, 1 個の外部流入出量は管網系全体の水収支の釣り合いより求めることにすると, n_1 個の連続式のうち 1 個の連続式は残りの (n_1-1) 個の連続式より導かれるので, 1 個の連続式は不用となる。通常基準節点水頭の節点における連続式を不用な連続式として省くので, 結局未知数は基準節点水頭以外の (n_1-1) 個の節点水頭であり, 一方, 方程式の数は, 基準節点水頭以外の (n_1-1) 個の節点での連続式となる。なお, 最終的に得られる (n_1-1) 個の方程式は, (n_1-1) 個の未知節点水頭に関する非線形方程式となるため直接解くことはできない。

2-2 計算式の誘導

管網計算においては, 管径, 管路長, 管粗度及びバルブ開度 (すなわちバルブ損失係数) は既知とし, 摩擦損失及びバルブ損失以外の損失は無視する。この場合, 管路流量 Q_{ij} は節点 i, j 間の損失水頭 $h_{ij} (= H_i - H_j)$ の非線形関数であるので, $Q_{ij}(h_{ij})$ を h_{ij} の近似値 h^*_{ij} (ただし $h^*_{ij} \neq 0$) のまわりに Taylor 展開し, 1 次の項まで取ると次式となる。

$$Q_{ij}(h_{ij}) = Q_{ij}(h^*_{ij}) + \left. \frac{\partial Q_{ij}}{\partial h_{ij}} \right|_{h_{ij}=h^*_{ij}} \Delta h_{ij} \dots\dots\dots(5)$$

ここに,

$$\Delta h_{ij} = h_{ij} - h^*_{ij} \dots\dots\dots(6)$$

ここで, Q_{ij} を近似する方法として Taylor 展開の高次の項までとる方法も考えられるが, 実際問題としては 1 次の項までとる 1 次近似法で十分であり, 2 次項以下を考慮する高次近似法を用いる利点は余りないようである⁸⁾。次に,

$$a_{ij} = \left. \frac{\partial Q_{ij}}{\partial h_{ij}} \right|_{h_{ij}=h^*_{ij}} \dots\dots\dots(7)$$

とすれば, 式(2)より

$$\frac{\partial h_{ij}}{\partial Q_{ij}} = \frac{1}{\alpha} r_{ij}^{-1/\alpha} |Q_{ij}|^{(1/\alpha)-1} + 2F_{vij} |Q_{ij}| \dots\dots\dots(8)$$

したがって,

$$a_{ij} = \left. \frac{\partial Q_{ij}}{\partial h_{ij}} \right|_{h_{ij}=h^*_{ij}} = \left\{ \left. \frac{\partial h_{ij}}{\partial Q_{ij}} \right|_{h_{ij}=h^*_{ij}} \right\}^{-1}$$

より,

$$a_{ij} = \{ (1/\alpha) r_{ij}^{-1/\alpha} |Q_{ij}(h^*_{ij})|^{(1/\alpha)-1} + 2F_{vij} |Q_{ij}(h^*_{ij})| \}^{-1} \dots\dots\dots(9)$$

また,

$$b_{ij} = Q_{ij}(h^*_{ij}) - a_{ij} h^*_{ij} \dots\dots\dots(10)$$

とすれば, 式(5)は次式となる。

$$Q_{ij}(h_{ij}) = Q_{ij}(h^*_{ij}) + a_{ij} \Delta h_{ij} \dots\dots\dots(11)$$

$$\text{または } Q_{ij} = a_{ij}(H_i - H_j) + b_{ij} \dots\dots\dots(12)$$

以上の Q_{ij} の接線近似法 (Newton 法) の関係を図-1 に示している。

式(12)を節点 i における流量連続式(1)に代入すれば,




図-1 管網計算における接線近似法の説明

$$\sum_j \{a_{ij}(H_i - H_j) + b_{ij}\} = -q_i - Q_{i0} \dots \dots \dots (13)$$

上式は各節点について成り立つので、各節点水頭に関する連立1次方程式が得られる。いま総節点数 n_1 個の管網において、節点1の水頭 H_1 を基準節点水頭とし、2-1 で述べたように基準節点水頭以外の $(n_1 - 1)$ 個の節点について式(13)をたてマトリックス表示すると次式となる。

$$A \cdot H = B \dots \dots \dots (14)$$

ここに、

A : $(n_1 - 1) \times (n_1 - 1)$ 行列で、その要素は a_{ij} で構成される。

H : $(n_1 - 1)$ ベクトルで、その要素は H_i を除く $(n_1 - 1)$ 個の節点水頭で構成される。

B : $(n_1 - 1)$ ベクトルで、その要素は $q_i, b_{ij}, a_{i1}, H_1, Q_{i0}$ で構成される。

ここで、式(14)の行列 A の i 行 j 列要素 a_{ij} 及びベクトル B の i 番目の要素 B_i には次の規則性がある。

$$A_{ii} = \sum_m a_{i+1,m} \quad (m \text{ は節点 } (i+1) \text{ の隣接節点番号}) \dots \dots \dots (15. a)$$

$$A_{ij} = -a_{i+1,j+1} \quad (i \neq j) \dots \dots \dots (15. b)$$

$$B_i = -q_{i+1} - \sum_m b_{i+1,m} + a_{i+1,1}H_1 + Q_{i-1,0} \dots \dots \dots (15. c)$$

ここに、

$$i = 1 \sim (n_1 - 1), \quad j = 1 \sim (n_1 - 1)$$

$a_{ij} = a_{ji}$ 、また、節点 i と節点 j が管路により接続されていない場合には $a_{ij} = b_{ij} = 0$

ここで用いる節点法は、各節点水頭を仮定して、これより得られる流量が、各節点での連続条件を満足するまで水頭の補正計算を繰り返すもので、まず h^*_{ij} を適当に与えて、式(9)、式(10)の $a_{ij}(h^*_{ij})$ 、 $b_{ij}(h^*_{ij})$ を計算し、式(14)の連立1次方程式を解き、各節点水頭 H を求める。次に、これより $h_{ij}(=H_i - H_j)$ を計算し、式(2)より各管路の流量 Q_{ij} を求める。この場合、 Q_{ij} には式(5)に示すように Taylor 展開の2次項以降の切り捨てによる線形化誤差が含まれているため、流量連続条件を満足しない。そこで、ここで求められた H から得られる h_{ij} を新たに次のステップの h^*_{ij} として同様の計算を行い、各節点での流量連続条件が許容値内に収まったところで計算を終了する。

2-3 $Q_{ij}(h^*_{ij})$ の計算法

H-W式の場合、式(2)の $\alpha = 0.54$ のため式(2)右辺第1項と第2項で Q_{ij} の次数が異なり、 $Q_{ij}(h_{ij})$ を陽に表現できない。このため式(2)より仮定損失水頭 h^*_{ij} に

対する流量 $Q_{ij}(h^*_{ij})$ を直接求めることはできないので、以下のように接線近似法により $Q_{ij}(h^*_{ij})$ を求めた後、式(9)、式(10)を用いて a_{ij} 、 b_{ij} を計算する。

まず、式(2)より h_{ij} は Q_{ij} の関数であるので、 $h_{ij} = h_{ij}(Q_{ij})$ と書け、 $Q_{ij} = Q^*_{ij}$ のときの h_{ij} が h^*_{ij} であるから次式が得られる。

$$h^*_{ij} = h_{ij}(Q^*_{ij}) \dots \dots \dots (16)$$

上式において Q^*_{ij} は、 Q^*_{ij} の第 n 回目の近似値 $Q_{ij}^{(n)}$ 及びその補正值 $\Delta Q_{ij}^{(n)}$ の和として次式で表されるとする。

$$Q^*_{ij} = Q_{ij}^{(n)} + \Delta Q_{ij}^{(n)} \dots \dots \dots (17)$$

よって、式(17)を式(16)に代入し、式(16)を $Q_{ij}^{(n)}$ のまわりに Taylor 展開し、1次の項までとると次式となる。

$$h^*_{ij} = h_{ij}(Q_{ij}^{(n)}) + \frac{\partial h_{ij}}{\partial Q_{ij}} \Big|_{Q_{ij}=Q_{ij}^{(n)}} \Delta Q_{ij}^{(n)} \dots \dots \dots (18)$$

式(18)より $\Delta Q_{ij}^{(n)}$ は、

$$\Delta Q_{ij}^{(n)} = \frac{h^*_{ij} - h_{ij}(Q_{ij}^{(n)})}{\partial h_{ij} / \partial Q_{ij} |_{Q_{ij}=Q_{ij}^{(n)}}} \dots \dots \dots (19)$$

となるので、式(2)、式(8)を上式に代入すると補正流量 $\Delta Q_{ij}^{(n)}$ は次式となる。

$$\Delta Q_{ij}^{(n)} = \frac{h^*_{ij} - r_{ij}^{-1/\alpha} |Q_{ij}^{(n)}|^{\alpha-1} Q_{ij}^{(n)} - F_{vij} |Q_{ij}^{(n)}| |Q_{ij}^{(n)}|}{(1/\alpha) r_{ij}^{-1/\alpha} |Q_{ij}^{(n)}|^{\alpha-1} + 2F_{vij} |Q_{ij}^{(n)}|} \dots \dots \dots (20)$$

よって、式(20)より補正流量 $\Delta Q_{ij}^{(n)}$ を求め、次式より $(n+1)$ 回目の Q^*_{ij} の近似値 $Q_{ij}^{(n+1)}$ が求まる。

$$Q_{ij}^{(n+1)} = Q_{ij}^{(n)} + \Delta Q_{ij}^{(n)} \dots \dots \dots (21)$$

後は $Q_{ij}^{(0)}$ すなわち初期管路流量を適当に仮定し、

$$|Q_{ij}^{(n+1)} - Q_{ij}^{(n)}| < \epsilon \dots \dots \dots (22)$$

ここに、 ϵ : 許容誤差範囲
を満足するまで反復した場合の $Q_{ij}^{(n+1)}$ を $Q_{ij}(h^*_{ij})$ とする。

ここで、初期管路流量 $Q_{ij}^{(0)}$ の与え方については、式(2)の右辺第1項の摩擦損失水頭もしくは第2項のバルブ損失水頭を無視して求めた Q_{ij} を $Q_{ij}^{(0)}$ にする方法などが考えられるが、本報では摩擦損失とバルブ損失の比を種々変化させた場合に、種々の $Q_{ij}^{(0)}$ を与えて計算反復回数を検討した結果を考慮して、次式のように式(2)において $\alpha = 1/2$ とおいて求めた Q_{ij} を $Q_{ij}^{(0)}$ とした。

$$Q_{ij}^{(0)} = (r_{ij}^{-2} + F_{vij})^{-1/2} |h^*_{ij}|^{-1/2} h^*_{ij} \dots \dots (23)$$

2-4 h^*_{ij} の設定法

以上により管網内にバルブが設置された場合にも節点数を増加させることなく節点法により管網計算を行う手法を示したが、2-2 で示した計算過程で用いる接線近似法では、初期仮定損失水頭 h^*_{ij} の近似度がよくないと過剰修正となり、なかなか収束しないことがある^{3),4)}。そこで、繰り返し回数を減少させ、計算時間を短くするための手法が多く提案されている⁴⁾ が、本報では初期仮定損失水頭 h^*_{ij} を次のようにして求める。

まず、各節点の水頭 H_i を管路両端の水頭差が 0 にならないように適当に与える。{例えば後述の 3. で示すプログラムでは、 $H_i = (\text{基準節点水頭 } H_1 - \text{節点番号 } i)$ としている。} この適当に与えた場合の水頭差を $h^*_{ij(0)}$ とおく。次に、式(2)を変形して、

$$Q_{ij} = (H_i - H_j) / (r_{ij}^{-1/\alpha} |Q_{ij}|^{1/\alpha - 1} + F_{vij} |Q_{ij}|) \dots\dots\dots (24)$$

となるので、 $h^*_{ij(0)}$ より求められる管路流量 $Q_{ij}(h^*_{ij(0)})$ を式(24)の右辺の Q_{ij} に用いて、式(24)を式(2)の形に表現すると、式(2)の a_{ij} 及び b_{ij} は以下ようになる。

$$a_{ij} = 1 / (r_{ij}^{-1/\alpha} |Q_{ij}(h^*_{ij(0)})|^{1/\alpha - 1} + F_{vij} |Q_{ij}(h^*_{ij(0)})|) \dots\dots\dots (25)$$

$$b_{ij} = 0 \dots\dots\dots (26)$$

この a_{ij} は、式(9)において $\alpha = 1$ 、 $F_{vij} |Q^*_{ij}|$ の係数を 1 としたものである。ここで、式(25)、式(26)の a_{ij} 、 b_{ij} を用いて式(3)の連立 1 次方程式を解き、式(4)から求めた H から算定した h_{ij} を $h^*_{ij(1)}$ とし、さらに、この $h^*_{ij(1)}$ を用いて同様の手順を n 回繰り返したものを $h^*_{ij(n)}$ とする。そして、この $h^*_{ij(n)}$ を 2-2 の接線近似法を用いる場合の損失水頭の近似値 h^*_{ij} とする。後述の表-1 のプログラムではこの繰り返し回数 n を 2 回としている。

3. 計算プログラム

表-1 に使用した計算プログラムを示している。使用言語は FORTRAN 77 である。本プログラムでは節点番号 1 を基準節点水頭にとるようにしており、各節点での流量連続式の許容誤差を $0.01(\text{m}^3/\text{h}) = 0.0028(\text{l}/\text{sec})$ 、式(22)の許容誤差 $\epsilon = 0.001(\text{m}^3/\text{h})$ としている。なお、本プログラムでは、式(4)の連立 1 次方程式の解法としては、直接解法であるガウスの消去法⁹⁾ を用いている。連立 1 次方程式の解法としては、これ以外にも小さい記憶容量ですむ反復近似法^{4),9)} や、式(4)の行列 A の要素には 0 が多いことを考慮して連立方程式の次元数を減少させた上で、直接法によって解く還元解法¹⁰⁾ などが提案されてい

る。

さて、バルブ損失係数 f_{vij} は、通常バルブの開度 θ (%) の関数として与えられるが、この関数値はバルブの種類や管径などによって異なる。福岡市では配水管に取り付けられている電動調整バルブの大部分はバタフライ弁であり、このバタフライ弁の θ と f_{vij} の関係は、福岡市水道局管網計算資料によると管径には無関係であり、その関数式は次式で表される。

$$f_{vij}(\theta) = \begin{cases} 165, 226 \times 10^{-0.18\theta} (0 < \theta < 12.5) \\ 3, 696 \times 10^{-0.06\theta} (12.5 \leq \theta < 45) \dots\dots\dots (27) \\ 221 \times 10^{-0.03\theta} (45 \leq \theta \leq 100) \end{cases}$$

本プログラムでは、簡単のためバルブ開度 θ (%) とバルブ損失係数 f_{vij} の関数式が式(27)で表される場合のみを考慮している。よって、実際の管網系において本プログラムを用いる場合には、その管網系に含まれるバルブの種類などに応じて別途 θ と f_{vij} の関係式をプログラムに組み込まなければならない。

なお、プログラム中では、バルブ開度 θ が 0.01% より小さい場合は $\theta = 0$ とみなし、このバルブの設置されている管路の流量 Q_{ij} は常に 0 として取り扱うので、この場合、式(3)の $a_{ij} = b_{ij} = 0$ となる。

本プログラムは総節点数 $n_1 = 100$ 、総管路数 $n_2 = 100$ 、外部流入出量数 $N = 10$ までの管網計算が行えるように配列をとっているが、配水管網が大きい場合は配列宣言文の次元を大きくすればよい。また、本プログラムにおいて解が求まりにくい場合は、前述の流量連続式及び式(2)の許容誤差の条件が厳しいことが考えられるので、この場合はこれらの許容誤差を大きくとればよい。

本プログラムの入力データは次のとおりである。

- (1) 総節点数 n_1 、総管路数 n_2 、外部流入出量数 N
- (2) 節点情報すなわち、
 - ① 節点番号 1 における基準節点水頭
 - ② $(N - 1)$ 個の外部から流入出する節点番号とその外部流入出量 (流出量をプラスとする) 及び N 個目の外部流入出量のある節点番号。 N 個目の節点における外部流入出量は管網系全体の水収支の釣り合いよりプログラムの中で求めるようにしている。
 - ③ 節点 i における節点取り出し水量 $q_i (i = 1 \sim n_1)$
- (3) 管路情報すなわち管路 i における両端の節点番号、流速係数 C_i 、管径 D_i 、管路長 L_i 、バルブ存在情報 IV_i (管路 i にバルブのある場合は $IV_i = 1$ 、ない場合は $IV_i = 0$) 及びバルブ開度 θ_i (%) ($IV_i = 0$ の場合は θ_i には適当な値、例えば 0 を与えておけばよい) ($i = 1 \sim n_2$)

表-1 計 算 プ ロ グ ラ ム

```

0001 C-----
0002 C----- PIPELINE NET ANALYSIS WITH VALVES -----
0003 C-----
0004 INTEGER NQOUT(10),K(100,2),IV(100)
0005 REAL*4 QOUT(10),QQ(100),C(100),D(100),L(100),THETA(100)
0006 REAL*8 H(100),Q(100),RIJ(100),AIJ(100),BIJ(100),AB(99,100)
0007 REAL*8 SH,DQ,SUMA,SUMB
0008 C----- DATA INPUT -----
0009 READ(5,*) N1,N2,N
0010 READ(5,*) H(1)
0011 READ(5,*) (NQOUT(I),QOUT(I),I=1,N-1)
0012 READ(5,*) NQOUT(N)
0013 READ(5,*) (QQ(I),I=1,N1)
0014 READ(5,*) (K(I,1),K(I,2),C(I),D(I),L(I),IV(I),THETA(I),I=1,N2)
0015 C----- N KOME NO GAIBU RYUSYUTURYO NO KEISAN -----
0016 SUM1=0.0
0017 SUM2=0.0
0018 DO 10 I=1,N1
0019 SUM1=SUM1+QQ(I)
0020 10 CONTINUE
0021 DO 11 I=1,N-1
0022 SUM2=SUM2+QOUT(I)
0023 11 CONTINUE
0024 QOUT(N)=-SUM1-SUM2
0025 C----- PIPE JOSU RIJ NO KEISAN -----
0026 ALFA=0.54
0027 DO 12 I=1,N2
0028 RIJ(I)=0.27853*C(I)*D(I)**2.63*L(I)**(-ALFA)
0029 12 CONTINUE
0030 C----- SYOKI KATEI SUITO NO INPUT -----
0031 DO 13 I=2,N1
0032 H(I)=H(1)-I
0033 13 CONTINUE
0034 C----- KATEI RYURYO QIJ(HIJ*) NO KEISAN -----
0035 G=9.8*3600**2
0036 PI=3.141592654
0037 EPS1=0.001
0038 SHMIN=1.0E-50
0039 ICOUNT=0
0040 1000 CONTINUE
0041 DO 14 I=1,N2
0042 SH=H(K(I,1))-H(K(I,2))
0043 IF (ABS(SH).LT.SHMIN) THEN
0044 WRITE(6,100) 'PIPE NO.',I,'NO SONSITU SUITO GA 0 DESU.'
0045 100 FORMAT(' ',2X,A,2X,I2,4X,A)
0046 STOP
0047 END IF
0048 IF (IV(I).EQ.0) THEN
0049 Q(I)=RIJ(I)*ABS(SH)**(ALFA-1.0)*SH
0050 GO TO 14
0051 END IF
0052 IF (THETA(I).LT.0.01) THEN
0053 Q(I)=0.0
0054 GO TO 14
0055 END IF
0056 FV=0.0*SFVIJ(THETA(I))/(G*PI**2*D(I)**4)
0057 Q(I)=(RIJ(I)**(-2)+FV)**(-0.5)*ABS(SH)**(-0.5)*SH
0058 1 CONTINUE
0059 HQ=RIJ(I)**(-1.0/ALFA)*ABS(Q(I))**(1.0/ALFA-1.0)*Q(I)
0060 +FV*ABS(Q(I))*Q(I)
0061 RHQ=(1.0/ALFA)*RIJ(I)**(-1.0/ALFA)*ABS(Q(I))**(1.0/ALFA-1.0)
0062 +2*FV*ABS(Q(I))
0063 DQ=(SH-HQ)/RHQ
0064 Q(I)=Q(I)+DQ
0065 IF (ABS(DQ).LT.EPS1) GO TO 14
0066 GO TO 1
0067 14 CONTINUE

```

```

0068 C----- RENZOKU_SIKI CHECK -----
0069     EPS2=0.01
0070     DO 15 I=1,N1
0071       CHECK=QQ(I)
0072       DO 16 J=1,N2
0073         IF(K(J,1).EQ.I) CHECK=CHECK+Q(J)
0074         IF(K(J,2).EQ.I) CHECK=CHECK-Q(J)
0075       16 CONTINUE
0076       DO 17 J=1,N
0077         IF(NQOUT(J).EQ.I) THEN
0078           CHECK=CHECK+QOUT(J)
0079           GO TO 2
0080         END IF
0081       17 CONTINUE
0082     2 CONTINUE
0083     IF(ABS(CHECK).GE.EPS2) GO TO 3
0084     15 CONTINUE
0085     GO TO 1100
0086     3 CONTINUE
0087 C----- KEISU AIJ,BIJ NO KEISAN -----
0088     DO 18 I=1,N2
0089       IF(IV(I).EQ.1.AND.THETA(I).LT.0.01) THEN
0090         AIJ(I)=0.0
0091         BIJ(I)=0.0
0092         GO TO 18
0093       END IF
0094       IF(IV(I).EQ.0) THEN
0095         FV=0.0
0096       ELSE
0097         FV=0.0*SFVIJ(THETA(I))/(G*PI**2*D(I)**4)
0098       END IF
0099 C --- HANPUKU-KINJI HO ---
0100     IF(ICOUNT.LE.1) THEN
0101       AIJ(I)=1.0/(RIJ(I)**(-1.0/ALFA)
0102     - *ABS(Q(I))**(1.0/ALFA-1.0)+FV*ABS(Q(I)))
0103       BIJ(I)=0.0
0104       GO TO 18
0105     END IF
0106 C --- SESSEN-KINJI HO ---
0107     AIJ(I)=1.0/(1.0/ALFA*RIJ(I)**(-1.0/ALFA)
0108     - *ABS(Q(I))**(1.0/ALFA-1.0)+2.0*FV*ABS(Q(I)))
0109     SH=H(K(I,1))-H(K(I,2))
0110     BIJ(I)=Q(I)-AIJ(I)*SH
0111     18 CONTINUE
0112 C----- MATRIX AB NO INPUT -----
0113     DO 19 I=1,N1-1
0114       DO 19 J=1,N1-1
0115         AB(I,J)=0.0
0116     19 CONTINUE
0117     DO 20 I=1,N1-1
0118       DO 20 J=1,N1-1
0119         IF(I.EQ.J) THEN
0120           SUMA=0.0
0121           DO 21 M=1,N2
0122             IF(I+1.EQ.K(M,1).OR.I+1.EQ.K(M,2)) SUMA=SUMA+AIJ(M)
0123           21 CONTINUE
0124           AB(I,J)=SUMA
0125         ELSE
0126           DO 22 M=1,N2
0127             IF((I+1.EQ.K(M,1).AND.J+1.EQ.K(M,2)).OR.
0128             - (I+1.EQ.K(M,2).AND.J+1.EQ.K(M,1))) AB(I,J)=-AIJ(M)
0129           22 CONTINUE
0130         END IF
0131     20 CONTINUE
0132     DO 23 I=1,N1-1
0133       SUMB=-QQ(I+1)
0134       DO 24 M=1,N2
0135         IF(I+1.EQ.K(M,1)) SUMB=SUMB-BIJ(M)
0136         IF(I+1.EQ.K(M,2)) SUMB=SUMB+BIJ(M)
0137         IF((I+1.EQ.K(M,1).AND.K(M,2).EQ.1).OR.
0138         - (I+1.EQ.K(M,2).AND.K(M,1).EQ.1)) SUMB=SUMB+AIJ(M)*H(I)
0139     24 CONTINUE

```

```

0140      DO 25 J=1,N
0141      IF(I+1.EQ.NQOUT(J)) THEN
0142          SUMB=SUMB-QOUT(J)
0143      GO TO 4
0144      END IF
0145      25 CONTINUE
0146      4 CONTINUE
0147      AB(I,N1)=SUMB
0148      23 CONTINUE
0149 C----- RENRITU 1JI HOTEISIKI NO KAIHO -----
0150      CALL GAUSS(AB,99,100,N1-1)
0151      DO 26 I=1,N1-1
0152          H(I+1)=AB(I,N1)
0153      26 CONTINUE
0154      ICOUNT=ICOUNT+1
0155      GO TO 1000
0156 C----- KEISAN KEKKA NO OUTPUT -----
0157      1100 CONTINUE
0158      WRITE(6,110) 'NUMBER OF NODAL POINTS      N1=',N1
0159      WRITE(6,110) 'NUMBER OF PIPES              N2=',N2
0160      WRITE(6,110) 'NUMBER OF EXTERNAL OUTFLOWS N=',N
0161      110 FORMAT(' ',2X,A,I3)
0162      WRITE(6,120) 'STANDARD WATER HEAD      H(1)=' ,H(1), '(M) '
0163      120 FORMAT('0',2X,A,F8.3,A)
0164      WRITE(6,130) 'EXTERNAL OUTFLOWS (M**3/HOUR) '
0165      130 FORMAT('0',2X,A)
0166      WRITE(6,140) ('QOUT(' ,NQOUT(I), ')=' ,QOUT(I), I=1,N)
0167      140 FORMAT(' ',5(3X,A,I3,A,F10.3))
0168      WRITE(6,150) 'NODAL', 'DEMAND', 'WATER',
0169      -           'NUMBER', '(M**3/H)', 'HEAD(M) '
0170      150 FORMAT('0',2X,A,4X,A,3X,A/' ',2X,A,2X,A,2X,A)
0171      DO 27 I=1,N1
0172          WRITE(6,160) I,QQ(I),H(I)
0173      160 FORMAT(' ',3X,I3,1X,F10.3,1X,F8.3)
0174      27 CONTINUE
0175      WRITE(6,170) 'PIPE', 'NODE NUMBER', 'COEF.C',
0176      -           'DIAMETER', 'LENGTH', 'IV(I)', 'VALVE', 'DISCHARGE'
0177      170 FORMAT('1',2X,A,4X,A,4X,A,4X,A,2X,A,2X,A,1X,A,1X,A)
0178      WRITE(6,180) 'NUMBER', 'K(I,1)', 'K(I,2)', '(M**0.37/H) ',
0179      -           'D (M)', 'L (M)', '(%)', '(M**3/H) '
0180      180 FORMAT(' ',2X,A,1X,A,1X,A,1X,A,4X,A,3X,A,10X,A,2X,A)
0181      DO 28 I=1,N2
0182          WRITE(6,190) I,K(I,1),K(I,2),C(I),D(I),L(I),IV(I),THETA(I),Q(I)
0183      190 FORMAT(' ',3X,I3,4X,I3,4X,I3,5X,F8.1,2X,F7.2,1X,
0184      -           F8.1,3X,I2,2X,F6.1,2X,F8.3)
0185      28 CONTINUE
0186      STOP
0187      END
0188 C-----C
0189 C--- VALVE SONSITU_KEISU NO KEISAN ---C
0190 C-----C
0191      FUNCTION SFVIJ(THETA)
0192      REAL*4 THETA,SFVIJ
0193      IF(THETA.GE.0.0.AND.THETA.LT.12.5)
0194      -   SFVIJ=165226*10**(-0.18*THETA)
0195      IF(THETA.GE.12.5.AND.THETA.LT.45.0)
0196      -   SFVIJ=3696*10**(-0.06*THETA)
0197      IF(THETA.GE.45.0.AND.THETA.LE.100.0)
0198      -   SFVIJ=221*10**(-0.03*THETA)
0199      RETURN
0200      END
0201 C-----C
0202 C--- GAUSS SYOKYO HO (RENRITU 1JI HOTEISIKI KAI) ---C
0203 C-----C
0204      SUBROUTINE GAUSS(A,NMAX,MMAX,N)
0205      REAL*8 A(NMAX,MMAX),TMAX,RMAX,EPS,H,P
0206      TMAX=DABS(A(1,1))
0207      DO 10 I=1,N
0208      DO 10 J=1,N
0209      IF(TMAX.LT.DABS(A(I,J))) TMAX=DABS(A(I,J))
0210      10 CONTINUE
0211      EPS=TMAX*1.0D-15

```

```

0212      DO 20 J=1,N-1
0213      RMAX=DABS(A(J,J))
0214      KW=J
0215      DO 30 I=J+1,N
0216          IF (DABS(A(I,J)).LE.RMAX) GO TO 30
0217          RMAX=DABS(A(I,J))
0218          KW=I
0219      30 CONTINUE
0220      IF (KW.EQ.J) GO TO 1
0221      DO 40 M=J,N+1
0222          W=A(J,M)
0223          A(J,M)=A(KW,M)
0224          A(KW,M)=W
0225      40 CONTINUE
0226      1 IF (DABS(A(J,J)).LT.EPS) THEN
0227          WRITE(6,100)
0228      100  FORMAT(1H,5X,'***** THIS MATRIX IS SINGULAR.
0229      -      CHECK THE MATRIX. *****')
0230          RETURN
0231      END IF
0232      P=1.0/A(J,J)
0233      DO 50 M=J+1,N+1
0234          A(J,M)=A(J,M)*P
0235      DO 50 I=J+1,N
0236          A(I,M)=A(I,M)-A(I,J)*A(J,M)
0237      50 CONTINUE
0238      20 CONTINUE
0239      IF (DABS(A(N,N)).LT.EPS) THEN
0240          WRITE(6,100)
0241          RETURN
0242      END IF
0243      A(N,N+1)=A(N,N+1)/A(N,N)
0244      DO 60 I=N-1,1,-1
0245      DO 60 J=I,1,-1
0246          A(J,N+1)=A(J,N+1)-A(J,I+1)*A(I+1,N+1)
0247      60 CONTINUE
0248      RETURN
0249      END

```

本プログラムは、管網内にバルブが設置されていない場合の管網計算にも容易に使用できる。また、例えばバルブのない管路に新たにバルブを取り付ける場合の結果を知りたいときなどにも、節点数を増加させることなく、ただ入力データのバルブ存在情報 IV_i を 1 にして、バルブ開度を与えるだけでよく、入力作業が非常に簡単である特長を有している。このことは模擬的にバルブの取り付けや取り外しを行い、管網の流量コントロールなどを検討するシミュレーションを行う場合には特に便利である。

以下にプログラムの主要な箇所について左の行番号にしたがって補足説明を加える。

- 9～14: 表-2のような形式による入力データの読み込み。
- 16～24: N 個目の外部流入出力量の計算。
- 26～29: 式(4)の r_{ij} の計算。
- 31～33: 初期仮定損失水頭 $h^*_{ij(0)}$ の設定。
- 35～67: $Q_{ij}(h^*_{ij})$ の計算。
- 69～86: 各節点における連続条件のチェック。
- 88～111: 各管路に対する式(3)の a_{ij} , b_{ij} の計算。

100～105: 接線近似法を用いる場合の損失水頭の近似値 h^*_{ij} を求めるため、 a_{ij} , b_{ij} を式(5), 式(6)より計算する。ここではこの過程を 2 回反復し、 $h^*_{ij(2)}$ を h^*_{ij} とする。

107～110: 式(9), 式(10)による a_{ij} , b_{ij} の計算。

113～148: 式(14)の行列 A 及びベクトル B からなる拡大係数行列⁹⁾ の計算。

119～124: 式(15. a)による A_{ii} の計算。

126～130: 式(15. b)による A_{ij} の計算。

132～148: 式(15. c)による B_i の計算。

150: ガウス消去法により、式(14)の連立 1 次方程式を解くサブルーチンの呼び出し。

引数 AB は行列 A 及びベクトル B からなる拡大係数行列で、この行列の左上 $\{(n_1-1) \times (n_1-1)\}$ の配列に式(14)の行列 A を入力し、また、 n_1 列目に式(14)の (n_1-1) 行ベクトル B を入力する。解ベクトル H は n_1 列目に (n_1-1) 行ベクトルとして出力される。

157～185: 表-3のような形式による結果の出力。

191～200: 式(27)によるバルブ開度 θ に対するバルブ損

表-2 入 力 デ ー タ

総節点数 n_1	総管路数 n_2	外部流入 出量数 N
29	37	3

基準節点水頭
 H_i (m)

30.0

節点 番号 i	外部流入出量 Q_0 (m^3/h)	管路 番号 i	管路両端の節点番号		流速係数 C_i ($m^{0.87}/h$)	管 径 D_i (m)	管 路 長 L_i (m)	バルブ情報 IV_i	バルブ開度 θ_i (%)
			$k(i,1)$	$k(i,2)$					
1	500.0	1	1	2	360,000	0.80	850	1	85
19	300.0	2	1	6	360,000	0.30	400	1	70
29		3	2	4	360,000	0.25	400	1	50
		4	2	5	360,000	0.80	600	0	0
		5	3	4	360,000	0.25	400	0	0
		6	4	13	360,000	0.25	800	0	0
		7	5	10	360,000	0.25	100	1	90
		8	6	7	360,000	0.20	800	0	0
		9	7	8	360,000	0.60	750	0	0
		10	7	14	360,000	0.60	1,100	1	15
		11	8	9	360,000	0.40	300	0	0
		12	8	15	360,000	0.80	200	0	0
		13	9	10	360,000	0.40	350	0	0
		14	9	12	360,000	0.20	200	0	0
		15	10	11	360,000	0.25	1,000	0	0
		16	10	13	360,000	0.25	100	0	0
		17	11	28	360,000	0.30	600	0	0
		18	12	13	360,000	0.25	500	0	0
		19	14	16	360,000	0.25	100	1	10
		20	14	18	360,000	0.70	850	0	0
		21	15	16	360,000	0.30	100	0	0
		22	16	20	360,000	0.30	1,100	1	35
		23	17	18	360,000	0.20	600	1	70
		24	18	19	360,000	0.70	150	0	0
		25	19	20	360,000	0.90	200	0	0
		26	20	21	360,000	0.90	100	0	0
		27	21	23	360,000	0.80	550	0	0
		28	22	23	360,000	0.30	1,200	1	80
		29	22	24	360,000	0.40	300	0	0
		30	23	25	360,000	0.80	400	0	0
		31	24	25	360,000	0.30	1,000	1	90
		32	24	26	360,000	0.35	200	0	0
		33	25	27	360,000	0.80	550	0	0
		34	26	27	360,000	0.40	600	1	25
		35	26	28	360,000	0.25	500	0	0
		36	27	29	360,000	0.60	350	0	0
		37	28	29	360,000	0.25	800	1	75

表-3 出力結果

NUMBER OF NODAL POINTS N1= 29
 NUMBER OF PIPES N2= 37
 NUMBER OF EXTERNAL OUTFLOWS N= 3


STANDARD WATER HEAD H(1)= 30.000(M)

EXTERNAL OUTFLOWS (M**3/HOUR)

QOUT(1)= -500.000 QOUT(19)= -300.000 QOUT(29)= 512.000

NODAL NUMBER	DEMAND (M**3/H)	WATER HEAD(M)
1	12.000	30.000
2	15.000	29.901
3	10.000	28.981
4	12.000	28.994
5	9.000	29.862
6	8.000	29.770
7	15.000	27.172
8	10.000	27.213
9	7.000	27.618
10	8.000	28.007
11	7.000	26.433
12	9.000	27.789
13	10.000	28.018
14	15.000	26.302
15	6.000	27.211
16	13.000	27.101
17	8.000	26.222
18	9.000	26.262
19	8.000	26.256
20	11.000	26.241
21	8.000	26.231
22	10.000	26.046
23	7.000	26.136
24	10.000	26.044
25	10.000	26.075
26	12.000	26.039
27	7.000	25.999
28	14.000	26.105
29	8.000	25.784

PIPE NUMBER	NODE NUMBER K(I,1)	NODE NUMBER K(I,2)	COEF.C (M**0.37/H)	DIAMETER D (M)	LENGTH L (M)	IV(I)	VALVE (%)	DISCHARGE (M**3/H)
1	1	2	360000.0	0.80	850.0	1	85.0	414.072
2	1	6	360000.0	0.30	400.0	1	70.0	73.928
3	2	4	360000.0	0.25	400.0	1	50.0	91.915
4	2	5	360000.0	0.80	600.0	0	0.0	307.156
5	3	4	360000.0	0.25	400.0	0	0.0	-10.000
6	4	13	360000.0	0.25	800.0	0	0.0	69.915
7	5	10	360000.0	0.25	100.0	1	90.0	298.156
8	6	7	360000.0	0.20	800.0	0	0.0	65.928
9	7	8	360000.0	0.60	750.0	0	0.0	-131.102
10	7	14	360000.0	0.60	1100.0	1	15.0	182.031
11	8	9	360000.0	0.40	300.0	0	0.0	-253.890
12	8	15	360000.0	0.80	200.0	0	0.0	112.781
13	9	10	360000.0	0.40	350.0	0	0.0	-228.776
14	9	12	360000.0	0.20	200.0	0	0.0	-32.115
15	10	11	360000.0	0.25	1000.0	0	0.0	80.181
16	10	13	360000.0	0.25	100.0	0	0.0	-18.801
17	11	28	360000.0	0.30	600.0	0	0.0	73.181
18	12	13	360000.0	0.25	500.0	0	0.0	-41.115
19	14	16	360000.0	0.25	100.0	1	10.0	-13.620
20	14	18	360000.0	0.70	850.0	0	0.0	180.650
21	15	16	360000.0	0.30	100.0	0	0.0	106.788
22	16	20	360000.0	0.30	1100.0	1	35.0	80.168
23	17	18	360000.0	0.20	600.0	1	70.0	-8.000
24	18	19	360000.0	0.70	150.0	0	0.0	163.650
25	19	20	360000.0	0.90	200.0	0	0.0	455.650
26	20	21	360000.0	0.90	100.0	0	0.0	524.819
27	21	23	360000.0	0.80	550.0	0	0.0	516.819
28	22	23	360000.0	0.30	1200.0	1	80.0	-24.944
29	22	24	360000.0	0.40	300.0	0	0.0	14.944
30	23	25	360000.0	0.80	400.0	0	0.0	484.874
31	24	25	360000.0	0.30	1000.0	1	90.0	-15.478
32	24	26	360000.0	0.35	200.0	0	0.0	20.422
33	25	27	360000.0	0.80	550.0	0	0.0	459.396
34	26	27	360000.0	0.40	600.0	1	25.0	29.466
35	26	28	360000.0	0.25	500.0	0	0.0	-21.044
36	27	29	360000.0	0.60	350.0	0	0.0	481.862
37	28	29	360000.0	0.25	800.0	1	75.0	38.138



①: 節点 i, j ; 管路 j , ◀▶: 圧力制御用バルブ, P: 水圧計, M: 流量計
 図-2 福岡市第9区画配水管網系

失係数 $f_{vij}(\theta)$ を計算する関数副プログラム。

204~249: ガウス消去法により連立1次方程式を解くサブルーチン副プログラム。

4. 適用例

本報では、図-2に示す福岡市第9区画配水管網系を例にとり管網計算を行った。図-2の管網系では総節点数 $n_1=29$, 総管路数 $n_2=37$, 外部からの流入量数 $N=3$ で、電動調整バルブ12個が設置されている。本適用例の場合の入力データを表-2に示している。この場合、管網計算の境界条件となる他の配水区画への外部流入量、ある節点のまわりの住民の需要量に相当する節点取り出し水量及びバルブ開度は表-2の Q_0, q_i, θ_i の欄のように今回の計算用に適当に与えて管網計算を行っている。表-3に、表-2の入力データに従って本プログラムにより管網計算を行った場合の出力結果を示している。表-3の WATER HEAD (M) の欄に各節点の水頭 (m) の値が、また、DISCHARGE (M³/H) の欄に各管路の流量 (m³/h) の値が管網計算の解として得られる。これらの解は、与えられた表-2の条件の下で、式(2)の水頭損失式の関係を満足し、そして、各節点での流量の連続式の誤差が 0.01(m³/h) 未満であることを示している。また、本プログラムではチェックのため入力データも表-3のように出力するようにしている。なお、使用計算機は九大計算機センターの FACOM M-780/20 であり、計算時間はプログラム翻訳などの時間も含めて0.44秒であった。また、連続条件を満足するまでの繰り返し回数は、接線近似法のための損失水頭の近似値を求める場合の反復も含めて7回であった。なお、 $h^{*ij(0)}$ を接線近似法の損失水頭の近似値 h^{*ij} に用いた場合には、繰り返し回数は41回で、計算時間は0.87秒を要した。

5. むすび

本論文では、配水管網のバルブによる圧力制御の問題

において必要不可欠となるバルブ損失水頭を考慮した節点法による配水管網計算法を、バルブの有無により計算節点数を変化させることなく計算する方法を定式化し、その計算プログラムを示した。本プログラムでは、管網内にバルブが設置されていない場合や任意の管にバルブを設置した場合でも、入力データの変更が非常に簡単である特長を有している。また、本プログラムはバルブ損失水頭を考慮した節点法による管網計算の基礎的な部分のみをプログラム化しており、また、節点番号1を基準節点水頭にとり、バルブ開度 $\theta(\%)$ とバルブ損失係数 f_{vij} の関数式が式(2)で表され、各節点での流量連続式の許容誤差を 0.01(m³/h)、式(2)の許容誤差 $\epsilon=0.001(\text{m}^3/\text{h})$ 、接線近似法の損失水頭の近似値 h^{*ij} には2-4で述べた反復回数を2回行った $h^{*ij(2)}$ を用いた条件の下でのプログラムである。よって、上記条件を変更する場合は、プログラム中の該当部分を変更する必要がある。また、連立1次方程式を解く場合などにおいて、省容量、省演算時間のためのプログラムが必要な場合は本プログラムを改良して使用していただければ幸いである。

終わりに、本研究において資料提供並びに種々の御協力をいただいた福岡市水道局水管理センター及び水道局計画課の方々には厚く御礼申し上げる。

参考文献

- 1) 大櫛星太: 水管理都市宣言, 水道協会雑誌, Vol. 52, No. 8, pp.16~22 (1983.8)
- 2) 河村 明・神野健二・上田年比古・土井敬介: 上水道配水管網系の節点需要量のオンライン予測に関する研究, 土木学会論文集, 第405号/II-11, pp.245~254 (1989.5)
- 3) 水理公式集, 土木学会編, pp.379~398 (1971.11)
- 4) 綾日出教: 配水施設のシミュレーション(I)—マイクロコンピュータによる管網計算一, 水道協会雑誌, No. 559, pp.29~42 (1981.4)
- 5) 高桑哲男: 配水管網流量計算法に関する研究(IV)—配水コントロールのための基礎的計算例一, 水道協会雑誌, No. 425, pp.32~47 (1970.2)
- 6) 高橋 澄・十時敏男: 電子計算機による配水管網の計算, 水道協会雑誌, No. 399, pp.41~46 (1967.12)
- 7) 高桑哲男: 配水管網流量計算法による研究(III)—節点エネルギー位を未知数とした計算法一, 水道協会雑誌, No. 423, pp.46~60 (1969.12)
- 8) 高桑哲男: 配水管網流量計算法に関する研究(I)—流量を未知数とした計算法と反復近似法による改良一, 水道協会雑誌, No. 421, pp.41~56 (1969.10)
- 9) 山下純一訳: アセトンのやさしい線形代数, 現代数学社 (1982.5)
- 10) 高桑哲男: 配水管網流量計算法に関する研究(II)—連立1次方程式の還元解法とその管網流量計算への応用一, 水道協会雑誌, No. 422, pp.27~36 (1969.11)

(平成元年3月31日受付)